
Teoría cinética molecular
El efecto de la temperatura y la masa en algunas propiedades que se pueden modelar con la teoría cinética 
molecular.

Distribución de Maxwell
Definiciones
Distribución de velocidades

La función G[v,T,M] describe la función de distribución de velocidades de Maxwell, mientras que F[v,T,M] repre-
senta la fracción de partículas con velocidades entre 0 y v.

G[v_,T_,M_]:=4 Pi v^2 (M/(2 Pi R T))^(3/2) Exp[-M v^2/(2 R T)];
F[v_,T_,M_]:=Erf[Sqrt[M v^2/(2 R T)]]- Sqrt[2 M v^2/(Pi R T)] Exp[-M v^2/(2 R T)];

La función Etr es la distribución de la energía cinética.

Etr[e_,T_]:=2 Pi/(Pi R T)^(3/2) Sqrt[e] Exp[-e/(R T)];

Las funciones g y f son las distribuciones para una componente de la velocidad.

g[v_,T_,M_]:=Sqrt[M/(2 Pi R T)] Exp[-M v^2/(2 R T)];
f[v_,T_,M_]:=Erf[Sqrt[M v^2/(2 R T)]];

Velocidades características: promedio, más probable y rms.

Vav[T_,M_]:=Sqrt[8 R T/(Pi M)];
Vmp[T_,M_]:=Sqrt[2 R T/M];
Vrms[T_,M_]:=Sqrt[3 R T/M];

Relación entre las velocidades características

r1=PowerExpand[Vrms[T,M]/Vav[T,M]];
r2=PowerExpand[Vrms[T,M]/Vmp[T,M]];
{{"Vrms/Vav=",N[r1]},{"Vrms/Vmp=",N[r2]}}//TableForm

Vrms/Vav= 1.0854
Vrms/Vmp= 1.22474

Constantes
Constante de los gases y masa molar de los gases nobles y el metano en unidades MKS.

R=8.314;

MHe=4.0026 10^-3;
MNe=20.1797 10^-3;
MAr=39.948 10^-3;
MKr=83.80 10^-3;
MXe=131.29 10^-3;

MCH4=16.035 10^-3;



Efecto de la temperatura
Distribución de velocidades del helio a 100, 300, 500, 700 y 900 K.
Observe el corrimiento del máximo de la distribución y el ensanchamiento.

To=100;
Plot[{G[v,To,MHe],G[v,3 To,MHe],G[v,5 To,MHe],G[v,7 To,MHe],G[v,9 To,MHe]}, 
{v,0,5000},PlotLegends->{To,3 To, 5 To,7 To, 9 To}]

1000 2000 3000 4000 5000

0.0002

0.0004

0.0006

0.0008

0.0010

0.0012

100

300

500

700

900

El cambio en el ancho de la distribución genera cambios en la distribución acumulada.

Plot[{F[v,To,MHe],F[v,3 To,MHe],F[v,5 To,MHe],F[v,7 To,MHe],F[v,9 To,MHe]}, 
{v,0,5000},PlotLegends->{To,3 To, 5 To,7 To, 9 To}]

Metano
Velocidades caracterísicas y distribución de velocidades a 300K.

To=400;
{{"Vmp","Vav","Vrms"}, {Vmp[To,MCH4],Vav[To,MCH4],Vrms[To,MCH4]}}//N//TableForm

vo={0,100,300,500,700,900,1100};
TableForm[Table[{vo[[i]], G[vo[[i]],To,MCH4], F[vo[[i]],To,MCH4]-F[vo[[i-1]],To,MCH4]}, 
{i,2,Length[vo]}]//N, TableHeadings->{{},{"v","G(v)","F(v)-F(v_)"}}]

Vmp Vav Vrms
644.044 726.726 788.79

v G(v) F(v)-F(v_)
100. 0.0000824647 0.00277552
300. 0.000611997 0.0640816
500. 0.00115591 0.181439
700. 0.00127027 0.251075
900. 0.000970824 0.228786
1100. 0.000552885 0.151884

Observe que el 75% de las partículas tienen velocidades entre 300 y 900 m s-1.

Plot[G[v,To,MCH4],{v,0,2000}]

Distribuciones de la velocidad y de una de sus componentes a 300K.

Plot[{G[v,To,MCH4],g[v,To,MCH4]}, {v,0,2000}]
Plot[{F[v,To,MCH4],f[v,To,MCH4]}, {v,0,2000}]

Efecto de la masa molar
Distribución de velocidades para los gases nobles a 300K.

To=300;
Plot[{G[v,To,MHe],G[v,To,MNe],G[v,To,MAr],G[v,To,MKr],G[v,To,MXe]}, {v,0,3000}, 
PlotRange->{0,0.005},PlotLegends->{"He","Ne","Ar","Kr","Xe"}]

Observe que el aumento en la masa disminuye el ancho de la distribución. este efecto es contrario al de la 
temperatura.

{{" ", "Vav"}, {"He",Vav[To,MHe]}, {"Ne",Vav[To,MNe]}, {"Ar",Vav[To,MAr]}, 
{"Kr",Vav[To,MKr]}, {"Xe",Vav[To,MXe]}}//N//TableForm

Vav
He 1259.69
Ne 561.02
Ar 398.739
Kr 275.305
Xe 219.948

Distribución de energía cinética
To=100;
Plot[{Etr[e,To],Etr[e,3 To],Etr[e,5 To],Etr[e,7 To],Etr[e,9 To]}, {e,0,10000},PlotRange->{0,0.0006},PlotLegends->{To,3 
To, 5 To,7 To, 9 To}]

Clear[R];
Print["Emp= ",e /. Solve[D[Etr[e,T],e]==0,e]];
Integrate[e^2 Etr[e,T],{e,0,Infinity}, Assumptions->{Re[R T]>0}]

